線形最小二乗法

種々の導出の具体例を示した．特に正規方程式での共分散行列，
特異値分解の導出や標準偏差との関係はNumRecipeを参照せよ．

コマンドleastsquareによるfitting

```
> restart:
  X:=[1,2,3,4]:
  Y:=[0,5,15,24]:
> with(plots):with(linalg):with(stats):
```

Warning, the name changecoords has been redefined
Warning, the protected names norm and trace have been redefined and unprotected

```
> l1:=pointplot(transpose([X,Y]),symbolsize=30):
> eq_fit:= fit[leastsquare[[x,y], y=a*x^2+b*x+c, {a,b,c}]]([X,
  Y]);
```

$$eq\_fit := y = x^2 + \frac{16}{5}\,x - \frac{9}{2}$$

```
> f1:=unapply(rhs(eq_fit),x);
```

$$f1 := x \rightarrow x^2 + \frac{16}{5}\,x - \frac{9}{2}$$

```
> p1:=plot(f1(x),x=0..4):
> display(p1,l1);
```



xi二乗の極小値から

```
> restart;
  X:=[1,2,3,4]:
  Y:=[0,5,15,24]:
  f1:=x->a*x^2+b*x+c:
> S:=0;
  for i from 1 to 4 do
    S:=S+(f1(X[i])-Y[i])^2;
  end do:
```

$$S := 0$$

```
> fS:=unapply(S,(a,b,c));
```

$$fS := (a, b, c) \rightarrow (a+b+c)^2 + (4\,a+2\,b+c-5)^2 + (9\,a+3\,b+c-15)^2 + (16\,a+4\,b+c-24)^2$$

```
> plot3d(subs(aa=1,fS(aa,b,c)),b=0..5,c=-10..10);
```




```
> eqs:={diff(expand(S),a)=0,
  diff(expand(S),b)=0,
  diff(expand(S),c)=0};
```

$$eqs := \{-1078 + 708\,a + 200\,b + 60\,c = 0, -302 + 200\,a + 60\,b + 20\,c = 0, -88 + 60\,a + 20\,b + 8\,c = 0\}$$

```
> solve(eqs,{a,b,c});
```

$$\left\{a = 1, b = \frac{16}{5}, c = -\frac{9}{2}\right\}$$

```
>
```

正規方程式(normal equations)

```
> restart;
  X:=[1,2,3,4]:
  Y:=[0,5,15,24]:
  f1:=x->a[1]+a[2]*x+a[3]*x^2:
> with(LinearAlgebra):
  Av:=Matrix(1..4,1..3):
> ff:=(x,i)->x^(i-1);
  for i from 1 to 3 do
  for j from 1 to 4 do
    Av[j,i]:=ff(X[j],i);
  end do;
  end do;
  Av;
```

$$ff := (x, i) \rightarrow x^{i-1}$$

$$\begin{bmatrix} 1 & 1 & 1 \\ 1 & 2 & 4 \\ 1 & 3 & 9 \\ 1 & 4 & 16 \end{bmatrix}$$

```
> Ai:=MatrixInverse(Transpose(Av).Av);
```

$$Ai := \begin{bmatrix} \frac{31}{4} & -\frac{27}{4} & \frac{5}{4} \\ -\frac{27}{4} & \frac{129}{20} & -\frac{5}{4} \\ \frac{5}{4} & -\frac{5}{4} & \frac{1}{4} \end{bmatrix}$$

```
> b:=Transpose(Av).Vector(Y);
```

$$b := \begin{bmatrix} 44 \\ 151 \\ 539 \end{bmatrix}$$

```
> Ai.b;
```

$$\begin{bmatrix} -\frac{9}{2} \\ \frac{16}{5} \\ 1 \end{bmatrix}$$

特異値分解(Singular Value Decomposition)

```
> restart;
  X:=[1,2,3,4]:
  Y:=[0,5,15,24]:
  f1:=x->a[1]+a[2]*x+a[3]*x^2:
> with(LinearAlgebra):
  Av:=Matrix(1..4,1..3):
> ff:=(x,i)->x^(i-1):
  for i from 1 to 3 do
  for j from 1 to 4 do
    Av[j,i]:=ff(X[j],i);
  end do;
  end do;
  Av;
```

$$\begin{bmatrix} 1 & 1 & 1 \\ 1 & 2 & 4 \\ 1 & 3 & 9 \\ 1 & 4 & 16 \end{bmatrix}$$

```
> U,S,Vt:=evalf(SingularValues(Av,output=['U','S','Vt'])):
> DiagonalMatrix(S[1..3],4,3);
  U.DiagonalMatrix(S[1..3],4,3).Vt:
```

$$\begin{bmatrix} 19.6213640200000015 & 0 & 0 \\ 0 & 1.71206987399999999 & 0 \\ 0 & 0 & 0.266252879300000022 \\ 0 & 0 & 0 \end{bmatrix}$$

```
> iS:=Vector(3):
  for i from 1 to 3 do
  iS[i]:=1/S[i];
  end do:
  DiS:=DiagonalMatrix(iS[1..3],3,4);
```

$$DiS := \begin{bmatrix} 0.05096485642 & 0 & 0 & 0 \\ 0 & 0.5840883104 & 0 & 0 \\ 0 & 0 & 3.755827928 & 0 \end{bmatrix}$$

```
> Transpose(Vt).DiS.(Transpose(U).Vector(Y));
```

$$\begin{bmatrix} -4.50000000198176498 \\ 3.20000000035008324 \\ 1.00000000040565196 \end{bmatrix}$$

```
>
```